
Abstract

This paper describes about article FUZZY 9. The article FUZZY 9, following on

from the previous article FUZZY 8, formulates various properties regarding triangular

membership functions. The main theorem is relationship between a triangular mem-

bership function composed of two straight lines and a MAX function and a triangular

membership function defined from the horizontal axis coordinates of the vertices of the

triangle. Moreover, defuzzified value of triangular membership function, construction of

trapezoidal membership function using triangular membership function and integration

of two connected functions are formalised.

1 はじめに
本アーティクルでは，[1]で定義されている TriangularFSとは異なる三角形型メンバ

シップ関数の表記を形式化した．前稿 [2]においては，絶対値関数を用いて二等辺三角形
型のメンバシップ関数の表記を形式化したが，本稿では左右非対称な関数について形式化
を行った．また，三角形型メンバシップ関数の重心法による非ファジィ化値（centroid [3]，
重心値）は三角形の高さに依存しないことを定理として表した．後半部には重心値計算に
まつわる区分的線形関数の積分に関するいくつかの性質を記述した．

図 1. Triangular (membership) functions

2 三角形型メンバシップ関数の諸性質
2.1 ２本の直線の方程式および最大値関数による三角型メンバシップ関

数の定義
[1]で定義されている三角形型メンバシップ関数（TriangularFS）は，以下の図 1の

三角形のように横軸の座標より次式のように定義されている．

TriangularFS(a, b, c)(x) =


0 x /∈ [a, c]
1

b− a
x− a

b− a
x ∈ [a, b]

− 1

c− b
x+

c

c− b
x ∈ [b, c]

(1)

この三角形型メンバシップ関数と，2本の直線および横軸によって三角形をなす三角
形型の関数とのパラメータの関連を定理としたものが，FUZZY 9:18および 9:19 である．
図 1の三角型メンバシップ関数の形式化は 9:18，図 2の三角型関数の形式化は 9:19であ
る．なお r = 1，pt− qs

p− s
= 1ならば通常のメンバシップ関数である．

図 2. A triangle surrounded by two lines and the horizontal axis

Listing 1. FUZZY 9：18, 19

theorem :: FUZZY 9:18
for a,b,c,r be Real, f be Function of REAL,REAL st a < b & b < c & r > 0 &
f = r (#) (AffineMap ( 1/(b − a), −(a/(b − a)) ) | ].−infty,b.] +∗ AffineMap ( −(1/(c − b)), c/(c − b) ) | [.b,+infty.[ )
holds
max+ f = r (#) TriangularFS (a,b,c);

theorem :: FUZZY 9:19
for p,q,s,t being Real, f being Function of REAL,REAL st p∗s < 0 & p > 0 & (p∗t−q∗s)/(p−s) > 0 &
for x being Real holds
f.x = ( (AffineMap (p,q) | ].−infty,(t−q)/(p−s).]) +∗ (AffineMap (s,t) | [.(t−q)/(p−s),+infty.[) ).x
holds
max+ f = (p∗t−q∗s)/(p−s) (#) ( TriangularFS(−q/p,(t−q)/(p−s),−t/s) );

上記 AffineMap [4]は以下のように一次関数を傾きと切片の定数で表している．

AffineMap(a, b)(x) = ax+ b (2)

また本研究ではmax+関数を用いた．これにより，メンバシップ関数値は 0以上であ
るので，関数値が 0になる区間の関数の表記が不要となる．max+関数はRFUNCT 3 [5]

で以下のとおり定義されている．

Listing 2. RFUNCT 3:def 1, 10

definition
let r be Real;
func max+ r −> Real equals :: RFUNCT 3:def 1
max (r,0);

definition
let D be non empty set ;
let F be PartFunc of D,REAL;
func max+ F −> PartFunc of D,REAL means :Def10: :: RFUNCT 3:def 10
( dom it = dom F & ( for d being Element of D st d in dom it holds it . d = max+ (F . d) ) );

このmax+関数を用いるために，いくつかの補題を用意した．以下はその一部である．

Listing 3. FUZZY 9：15, 16

theorem :: FUZZY 9:15
for a,b,c,x be Real st a < b & b < c & (not x in ].a,c.[) holds
( AffineMap ( 1/(b − a), −(a/(b − a)) ) | ].−infty,b.] +∗ AffineMap ( −(1/(c − b)) , c/(c − b) ) | [.b,+infty.[ ).x <= 0;

theorem :: FUZZY 9:16
for a,b,c,x be Real st a < b & b < c & x in [.a,c.] holds TriangularFS (a,b,c).x >= 0;

FUZZY 9:19の以下の仮定

ps < 0, p > 0,
pt− qs

p− s
> 0 (3)

より，TriangularFSの引数の条件

−q

p
<

t− q

p− s
,
t− q

p− s
< − t

s
(4)

が次のように導かれる．上の式において，右辺から左辺を引くと
t− q

p− s
− (−q

p
) =

p(t− q) + q(p− s)

p(p− s)
=

pt− qs

p(p− s)
> 0

− t

s
− t− q

p− s
= − pt− qs

s(p− s)
> 0 (5)

が得られる．ただし，式 (5)は s < 0より成立する．3つの不等式 (3)より (4)が導かれ
る命題は FUZZY 9:9に収録した．

Listing 4. FUZZY 9：9

theorem :: FUZZY 9:9
for p,q,s,t being Real st p∗s < 0 & p > 0 & (p∗t−q∗s)/(p−s) > 0 holds
−q/p < (t−q)/(p−s) & (t−q)/(p−s) < −t/s & −q/p < −t/s;

2.2 三角形型メンバシップ関数の面積と重心
次の定理 9:20は，図 1の (a,0),(b,r),(c,0)の三角型の面積が r(c− a)

2
であることを示

し，9:22は TriangularFS(a,b,c)を r倍した関数の centroidが三角形型メンバシップ関数
の高さに依らず a+ b+ c

3
であることを示している [6]．

Listing 5. FUZZY 9：20, 22

theorem :: FUZZY 9:20
for a,b,c,r be Real, f be Function of REAL,REAL st a < b & b < c &
for x being Real holds f.x = (r (#) TriangularFS (a,b,c)).x
holds
integral(f,[’a,c’]) = r∗(c−a)/2;

theorem :: FUZZY 9:22
for a,b,c,r be Real, f be Function of REAL,REAL st a < b & b < c & r <> 0 &
for x being Real holds f.x = (r (#) TriangularFS (a,b,c)).x
holds
centroid(f,[’a,c’]) = (a+b+c)/3;

2.3 三角形型メンバシップ関数のその他の数学的性質
筆者らはファジィ推論計算の連続性を用いてファジィ制御の最適フィードバックの存

在性に関する考察を行ってきた [7]．以下の定理 9:24の Lipschitz連続性は状態方程式の
解の一意存在の証明に利用でき，9:25は区分的線形関数による三角形型（メンバシップ）
関数の有界性および積分可能性を表し，centroidの計算に必要な性質の一部である．

Listing 6. FUZZY 9：24, 25

theorem :: FUZZY 9:24
for A be non empty closed interval Subset of REAL,
p,q,r,s,t,u be Real, f be Function of REAL,REAL st p∗s < 0 & p > 0 &
for x be Real holds
f.x = max(r,min( u, ( (AffineMap (p,q) |].−infty,(t−q)/(p−s).]) +∗ (AffineMap (s,t) |[.(t−q)/(p−s),+infty.[) ) .x ))
holds
f is Lipschitzian;

theorem :: FUZZY 9:25
for A be non empty closed interval Subset of REAL,
p,q,r,s,t,u be Real, f be Function of REAL,REAL st p∗s < 0 & p > 0 &
for x be Real holds
f.x = max(r,min( u, ( (AffineMap (p,q) |].−infty,(t−q)/(p−s).]) +∗ (AffineMap (s,t) |[.(t−q)/(p−s),+infty.[) ) .x ))
holds
f is integrable on A & f | A is bounded;

FUZZY 9：24は，FUZZY 6:28，FUZZY 7:42より得られる．FUZZY 9:24の関数 f

を
f(x) = max{r,min{u, g(x)}} (6)

ただし，

g(x) =


px+ q x ≤ t− q

p− s

sx+ t x ≥ t− q

p− s

(7)

とする．FUZZY 6:27より AffineMap(p, q).x = px + q の Lipschitz定数の一つとして
|p|+1が存在する．一方，FUZZY 6:26より Lipschitz連続な関数が演算+∗で結合した関

数の Lipschitz定数はそれらの最大値が一つとして存在する．したがって，f の Lipschitz

定数として (max{p,−s}+ 1)が存在して，任意の x1, x2 ∈ Rに対して，

|f(x1)− f(x2)| ≤ (max{p,−s}+ 1) |x1 − x2|

が成立する．

2.4 ２つの三角形型メンバシップ関数による台形型メンバシップ関数の
構成

図 3の台形型メンバシップ関数は２つの三角形型メンバシップ関数の和算によって形
成ができる．これを表すのが FUZZY 9:34である．代数積-加算-重心法 [8] による後件部
のメンバシップ関数の加算による合成の特別な場合である．

図 3. Trapezoidal membership function, TrapezoidalFS (a,b,c,d)

Listing 7. FUZZY 9：34

theorem :: FUZZY 9:34
for a, b, c, d being Real st a < b & b < c & c < d holds
TriangularFS (a,b,c) + TriangularFS (b,c,d) = TrapezoidalFS (a,b,c,d);

3 連結する２つの関数の性質
本節では，２つの関数が演算 +∗で結合した関数の性質について，その一部を紹介す

る．以下はスカラ倍に関する定理である．

Listing 8. FUZZY 9：1, 2

theorem :: FUZZY 9:1
for a,b,r be Real, f be Function of REAL,REAL holds
r (#) (f | [.a,b.]) = (r (#) f) | [.a,b.];

theorem :: FUZZY 9:2
for r,c,d,u,v be Real, f,g be Function of REAL,REAL holds
r (#) ((f | [.c,d.]) +∗ (g | [.u,v.])) = (r (#) (f | [.c,d.])) +∗ (r (#) (g | [.u,v.]));

本稿では centroid計算に必要な積分に関するいくつかの定理を証明した．次の定理は
FUZZY 8:14および 16 [2]の被積分関数の定義域の表現を変えたものとなっている．

Listing 9. FUZZY 9：38, 39

reserve A for non empty closed interval Subset of REAL;

theorem :: FUZZY 9:38 ::
for a being Real, f, g, h being Function of REAL,REAL st
f | A is bounded & f is integrable on A & g | A is bounded & g is integrable on A &
a in A & h = ( f | ].−infty,a.] ) +∗ ( g | [.a,+infty.[ ) & f.a = g.a
holds
h is integrable on [’ lower bound A, a ’] & h is integrable on [’ a, upper bound A ’];

theorem :: FUZZY 9:39
for a being Real, f, g, h being Function of REAL,REAL st
f | A is bounded & f is integrable on A & g | A is bounded & g is integrable on A &
a in A & h = ( f | ].−infty,a.] ) +∗ ( g | [.a,+infty.[ ) & f.a = g.a
holds
integral (h,A) = integral (f,[’lower bound A, a’]) + integral (g,[’a,upper bound A’]);

有界閉区間Aにおいて有界かつ積分可能な関数 f および gが a ∈ Aで下の式 (8)のよ
うに連結している関数を hとするとき，hの積分は以下の式で表される．

h(x) =


f(x) x ≤ a

f(a) = g(a) x = a

g(x) x ≥ a

(8)

∫
A

h(x)dx =

∫ a

inf A

f(x)dx+

∫ supA

a

g(x)dx

この証明には hが定義域，値域ともに REALの Functionである必要があるため以下
の定理 [9]を適用した．

Listing 10. FUZZY 7：12

theorem :: FUZZY 7:12
for s be Real, f,g being Function of REAL,REAL holds
( f | ].−infty,s.] ) +∗ ( g | [.s,+infty.[ ) is Function of REAL,REAL;

さらに，f および gが連続であるとき，次が成立する．

centroid(h, [′d, e′]) =

∫ e

d
xh(x)dx∫ d

e
h(x)dx

=

∫ a

d
xf(x)dx+

∫ e

a
xg(x)dx∫ d

e
h(x)dx

この式の定式化は以下の定理である．

Listing 11. FUZZY 9:3

theorem :: FUZZY 9:3
for f,g,h be Function of REAL,REAL, a,d,e be Real st
d <= a & a <= e & f is continuous & g is continuous &
h | [.a,c.] = f | [.a,b.] +∗ g | [.b,c.] &
integral(f,[’d,a’]) <> 0 & integral(g,[’a,e’]) <> 0 & f.a = g.a holds
centroid(h, [’d,e’]) = 1 / integral(h,[’d,e’])
∗ ( integral((id REAL)(#)f,[’d,a’]) + integral((id REAL)(#)g,[’a,e’]) );

ただし，定数 d, a, eは上記の式 (8) に合わせて変更してある．

4 まとめ
本研究では，２本の直線（AffineMap）によって形成される三角形型メンバシップ関

数について関数 TriangularFSとの一致・変換を定理として形式化を行った．関数max+

を用いることにより，区分的線形関数は２つの直線の方程式だけで形成され簡略化された
表記であると考える．本アーティクルに記述したメンバシップ関数は前件部よりはむしろ
後件部の構成に有用であると期待できる．次のアーティクルでは台形型のメンバシップ関
数について同様の定理類を形式化することを予定している．

